

**Diese Anleitung nur benutzen, wenn die Datei Internet Explorer 7.zip nicht verwendet wird.
Sonst die Datei readME.txt lesen**
#####

Test Internet Explorer 7 ohne zu installieren

Um Internet Explorer 7 zu installieren, brauchst du Windows XP SP2.

Schritt 1

Download Internet Explorer 7 Beta 2 (11.9 MB). Diese Anleitung funktioniert nicht mit IE7 Beta 3.

Schritt 2

Mit WinRAR (download Testversion) Dateien von der Setup-Datei in den Ordner - C:\Programme\Internet Explorer 7 extrahieren.



Es geht auch ohne WinRAR. Ein DOS-Fenster öffnen, mit cd zum Ordner mit der exe-Datei wechseln und "IE7BETA2-WindowsXP-x86-enu.exe /X" (richtiger Dateiname abhängig von der Downloaddatei) eingeben.

Schritt 3

Neue Datei ie7.bat im selben Ordner erstellen und den Code unten kopieren (courtesy of [Jon Galloway](#)):

```
@ECHO OFF
TITLE IE7 Launcher 1.4
ECHO IE7 STANDALONE LAUNCHER 1.4
ECHO Updated for IE7 Beta 2 Preview
ECHO.
ECHO Do not close this window or it will not clean up after itself properly.
ECHO You can pass a URL into this batch file, like this:
ECHO ie7.bat www.microsoft.com
ECHO.
ECHO When you close IE7, this will remove the registry key and shut itself down.
ECHO.
ECHO Setting up IE7 for standalone mode...
PUSHD %~dp0
ECHO Removing IE7 registry key and set the version vector to "7.0000".
> %TEMP%\IE7Fix.reg ECHO REGEDIT4
>>%TEMP%\IE7Fix.reg ECHO.
>>%TEMP%\IE7Fix.reg ECHO [-HKEY_CLASSES_ROOT\CLSID\{C90250F3-4D7D-4991-9B69-A5C5BC1C2AE6}]
>>%TEMP%\IE7Fix.reg ECHO [-HKEY_CLASSES_ROOT\Interface\{000214E5-0000-0000-C000-000000000046}]
>>%TEMP%\IE7Fix.reg ECHO [HKEY_LOCAL_MACHINE\SOFTWARE\Microsoft\Internet Explorer\Version Vector]
>>%TEMP%\IE7Fix.reg ECHO "IE"="7.0000"
>>%TEMP%\IE7Fix.reg ECHO.
:: Merge the REG file to delete the IE7 standalone entry
REGEDIT /S %TEMP%\IE7Fix.reg
REN SHLWAPI.DLL SHLWAPI.DLL.BAK
TYPE NUL > IEXPLORE.exe.local
ECHO Running IE7...
iexplore.exe "%1"
:: Merge the REG file to delete the IE7 standalone entry
REGEDIT /S %TEMP%\IE7Fix.reg
:: Delete the temporary REG file
DEL %TEMP%\IE7Fix.reg
ECHO Removing IE7 standalone files...
REN SHLWAPI.DLL.BAK SHLWAPI.DLL
DEL IEXPLORE.exe.local
:: Set the old version vector "6.0000".
> %TEMP%\IE7Fix.reg ECHO REGEDIT4
>>%TEMP%\IE7Fix.reg ECHO.
>>%TEMP%\IE7Fix.reg ECHO [HKEY_LOCAL_MACHINE\SOFTWARE\Microsoft\Internet Explorer\Version Vector]
>>%TEMP%\IE7Fix.reg ECHO "IE"="6.0000"
>>%TEMP%\IE7Fix.reg ECHO.
REGEDIT /S %TEMP%\IE7Fix.reg
```

```
DEL %TEMP%\IE7Fix.reg
POPD
ECHO Complete, closing...
```

Schritt 4

Im gleichen Ordner eine neue Datei **ie7.vbs** mit diesem Code erstellen:

```
Set oShell = CreateObject("WScript.Shell")
oShell.Run "ie7.bat", 0, True
```

Schritt 5

Verknüpfung zu **ie7.vbs** auf dem Desktop erstellen. Icon der Verknüpfung mit dem vom IE7 wechseln.

Schritt 6

Im Unterordner **update** **xmllitesetup.exe** ausführen. (IE7 hat dann Tabs)

Jetzt hast du einen Standalone Internet Explorer 7 der neben dem Internet Explorer 6 läuft. IE7 mit der Verknüpfung auf dem Desktop starten.

Quelle: <http://googlesystem.blogspot.com/2006/04/test-internet-explorer-7-without.html>